
Change 1 

	Function name
	fl_TC_7_2_2_7_TestBody

	Reason for change
	The TC 7.2.2.7 has SDU sizes that exceed the limitation for UL/DL grant for Cat-M1 UEs. The SDU sizes and grant were modified to fit in 1000 bit TBS for Cat-M1 UEs.

	Summary of change
	Modified the SDU sizes and UL grant in step 6 for Cat-M1 UEs.

	TTCN module
	LTE\7_2\RLC_UM_Testcases.ttcn

	MCC160 Comment
	


Before change

    function fl_TC_7_2_2_7_TestBody(inout RLC_SS_State p_RLC_Rec) runs on EUTRA_PTC

  { // @sic R5s100006 sic@, @sic R5s100140 change 2 sic@ @sic R5-105016 sic@

    var integer v_RLCSDUsize := 128;

    var SubFrameTiming_Type v_Timing;

    var integer v_Subframe := 0;

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

    // determine RAT to make the timing correct later on

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    fl_GenerateRLC_UM_SDUs (p_RLC_Rec, v_RLCSDUsize, 4);

    //@siclog "Step 1" siclog@

    // The SS transmits an UMD PDU. This PDU carries SDU#1. SN equals 0.

    f_TxUMD_oneSDU( p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU1 );

    //@siclog "Step 2" siclog@

    // Does the UE transmit RLC SDU#1?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU1 );

    //@siclog "Step 2A" siclog@

    // The SS stops UL grant transmission

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync (eutra_Cell1, cs_TimingInfo_Now);

    //@siclog "Step 3" siclog@

    // The SS transmits an UMD PDU. This PDU contains the last part of SDU#4. SN equals 3.

    p_RLC_Rec.UM_VTUS  := 3;

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    v_Subframe := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    f_TxUMD_PDUpartofSDU (p_RLC_Rec,

                          cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_Subframe, v_Timing.HSFN.Number),

                          tsc_RLC_SDU4, 75, v_RLCSDUsize - 75,

                          tsc_FI_EndOfSDU);

    //@siclog "Step 4" siclog@

    // The SS transmits an UMD PDU. This PDU contains the last part of SDU#3, and the 1st part of SDU#4. SN equals 2.

    p_RLC_Rec.UM_VTUS  := 2;

    v_Subframe := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

    f_TxUMD_PDUpartofSDU12 (p_RLC_Rec,

                            cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_Subframe, v_Timing.HSFN.Number),

                            tsc_RLC_SDU3, 75, v_RLCSDUsize - 75,

                            tsc_RLC_SDU4, 0, 75,

                            tsc_FI_MiddleOfSDU);

    //@siclog "Step 5" siclog@

    // The SS transmits an UMD PDU. This PDU carries SDU#2 and the 1st part of SDU#3. SN equals 1.

    p_RLC_Rec.UM_VTUS  := 1;

    v_Subframe := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

    f_TxUMD_PDUpartofSDU12 (p_RLC_Rec,

                            cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_Subframe, v_Timing.HSFN.Number),

                            tsc_RLC_SDU2, 0, v_RLCSDUsize,

                            tsc_RLC_SDU3, 0, 75,

                            tsc_FI_StartOfSDU );

    //@siclog "Step 5A" siclog@

    // The SS waits for 60 ms and then allocates 3 UL grants (UL grant allocation type 2)(of size 1064 bits) for the UE

    // to loop back one SDU in one UMD PDU.

    // v_Timing.Subframe.Number of 4 is valid for UL grant transmission

    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);

    f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1,

                                               cs_TimingInfo(v_Timing),

                                               4, 15, 10, 3);  // Imcs = 4 -> Itbs = 4  @sic R5s110394, R5-156154 sic@

    //@siclog "Step 6" siclog@

    // Check: Does the UE transmit RLC SDU#2?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU2 );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Step 7" siclog@

    // Check: Does the UE transmit RLC SDU#3?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU3 );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");

    //@siclog "Step 8" siclog@

    // Check: Does the UE transmit RLC SDU#4?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU4 );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

  } // fl_TC_7_2_2_7_TestBody

After change

 import from Parameters all;
 function fl_TC_7_2_2_7_TestBody(inout RLC_SS_State p_RLC_Rec) runs on EUTRA_PTC

  { // @sic R5s100006 sic@, @sic R5s100140 change 2 sic@ @sic R5-105016 sic@

    var integer v_RLCSDUsize := 128;   

    var SubFrameTiming_Type v_Timing;

    var integer v_Subframe := 0;

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

    var integer v_Part1size := 75;    

    var integer v_Imcs := 4;

    var integer v_Nprb := 15;
    // determine RAT to make the timing correct later on

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    if (pc_ue_CategoryDL_M1){

      v_RLCSDUsize := 80;

      v_Part1size := 30;

      v_Imcs := 8;

      v_Nprb := 5;

    }
    fl_GenerateRLC_UM_SDUs (p_RLC_Rec, v_RLCSDUsize, 4);

    //@siclog "Step 1" siclog@

    // The SS transmits an UMD PDU. This PDU carries SDU#1. SN equals 0.

    f_TxUMD_oneSDU( p_RLC_Rec, cs_TimingInfo_Now, tsc_RLC_SDU1 );

    //@siclog "Step 2" siclog@

    // Does the UE transmit RLC SDU#1?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU1 );

    //@siclog "Step 2A" siclog@

    // The SS stops UL grant transmission

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync (eutra_Cell1, cs_TimingInfo_Now);

    //@siclog "Step 3" siclog@

    // The SS transmits an UMD PDU. This PDU contains the last part of SDU#4. SN equals 3.

    p_RLC_Rec.UM_VTUS  := 3;

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    v_Subframe := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    f_TxUMD_PDUpartofSDU (p_RLC_Rec,

                          cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_Subframe, v_Timing.HSFN.Number),

                          tsc_RLC_SDU4, v_Part1size, v_RLCSDUsize - v_Part1size,  

                          tsc_FI_EndOfSDU);

    //@siclog "Step 4" siclog@

    // The SS transmits an UMD PDU. This PDU contains the last part of SDU#3, and the 1st part of SDU#4. SN equals 2.

    p_RLC_Rec.UM_VTUS  := 2;

    v_Subframe := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

    f_TxUMD_PDUpartofSDU12 (p_RLC_Rec,

                            cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_Subframe, v_Timing.HSFN.Number),

                            tsc_RLC_SDU3, v_Part1size, v_RLCSDUsize - v_Part1size, 

                            tsc_RLC_SDU4, 0, v_Part1size,

                            tsc_FI_MiddleOfSDU);

    //@siclog "Step 5" siclog@

    // The SS transmits an UMD PDU. This PDU carries SDU#2 and the 1st part of SDU#3. SN equals 1.

    p_RLC_Rec.UM_VTUS  := 1;

    v_Subframe := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

    f_TxUMD_PDUpartofSDU12 (p_RLC_Rec,

                            cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_Subframe, v_Timing.HSFN.Number),

                            tsc_RLC_SDU2, 0, v_RLCSDUsize,

                            tsc_RLC_SDU3, 0, v_Part1size,                             

                            tsc_FI_StartOfSDU );

    //@siclog "Step 5A" siclog@

    // The SS waits for 60 ms and then allocates 3 UL grants (UL grant allocation type 2)(of size 1064 bits) for the UE

    // to loop back one SDU in one UMD PDU.

    // v_Timing.Subframe.Number of 4 is valid for UL grant transmission

    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);

    f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1,

                                               cs_TimingInfo(v_Timing),

                                               v_Imcs, v_Nprb, 10, 3);  // Imcs = 4 -> Itbs = 4  @sic R5s110394, R5-156154 sic@       //@siclog "Step 6" siclog@

    // Check: Does the UE transmit RLC SDU#2?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU2 );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Step 7" siclog@

    // Check: Does the UE transmit RLC SDU#3?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU3 );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");

    //@siclog "Step 8" siclog@

    // Check: Does the UE transmit RLC SDU#4?

    f_RxUMD_oneSDU( p_RLC_Rec, tsc_RLC_SDU4 );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

  } // fl_TC_7_2_2_7_TestBody
Change 2
	Function name
	fl_TC_7_2_2_8_TestBody

	Reason for change
	The TC 7.2.2.8 has SDU sizes that exceed the limitation for UL/DL grant for Cat-M1 UEs. The SDU sizes were modified to fit in 1000 bit TBS for Cat-M1 UEs.

	Summary of change
	Modified the SDU sizes for Cat-M1 UEs.

	TTCN module
	LTE\7_2\RLC_UM_Testcases.ttcn

	MCC160 Comment
	


Before change

function fl_TC_7_2_2_8_TestBody(inout RLC_SS_State p_RLC_Rec) runs on EUTRA_PTC

  {

    var integer v_RLC_SDUsize := 128;

    var DRB_COMMON_IND v_ReceivedAsp;

    var RLC_DataField_Type v_SDU2_Received;

    var RLC_DataField_Type v_RLC_Data1;

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

    var SubFrameTiming_Type v_Timing, v_SubFrameTimingT1, v_SubFrameTimingT2;

    var integer v_SubframeOffset := 0;

    var integer v_Duration;

    timer t_WatchDog := 0.1;

    // determine RAT to make the timing correct later on

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    fl_GenerateRLC_UM_SDUs (p_RLC_Rec, v_RLC_SDUsize, 2);

    v_RLC_Data1 := substr(p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2],2,(v_RLC_SDUsize-2));

    v_SDU2_Received := f_PDCP_PDU_Encvalue (cs_PDCP_DataLongSN(0,v_RLC_Data1));

    //@siclog "Step 1" siclog@

    // The SS transmits RLC PDU#1 containing first segment of RLC SDU#1.

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    // Note T1

    v_SubFrameTimingT1.HSFN.Number := v_Timing.HSFN.Number;

    v_SubFrameTimingT1.SFN.Number := v_Timing.SFN.Number;

    v_SubFrameTimingT1.Subframe.Number := v_SubframeOffset;

    f_TxUMD_PDUpartofSDU (p_RLC_Rec,

                          cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number),

                          tsc_RLC_SDU1, 0, 75, tsc_FI_StartOfSDU);

    //@siclog "Step 2" siclog@

    // The SS does not transmit RLC PDU#2 containing last segment of RLC SDU#1.

    // nothing sent in next subframe

    //@siclog "Step 3" siclog@

    // The SS transmits RLC PDU#3 containing RLC SDU#2.

    p_RLC_Rec.UM_VTUS  := 2;

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

    f_TxUMD_oneSDU( p_RLC_Rec,

                    cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number),

                    tsc_RLC_SDU2 );

    //@siclog "Step 3A" siclog@

    // Check 1: Does the UE transmit RLC SDU#2 after t-Reordering (200 ms) expiry?

    // Note T2.

    DRB.receive (car_DRB2_RLCUmDataInd (eutra_Cell1,

                                        cr_UMD_PDUoneSDU_SN10(p_RLC_Rec.UM_VRUR,

                                                              {v_SDU2_Received} ))) -> value v_ReceivedAsp;

    // Note T2.

    v_SubFrameTimingT2 := v_ReceivedAsp.Common.TimingInfo.SubFrame;

    v_Duration := f_EUTRA_NB_SubFrameTimingDuration( v_SubFrameTimingT1, v_SubFrameTimingT2);

    // Check 2: Is (T2-T1) > t-Reordering?

    if ( f_EUTRA_SubFrameTimingCheckDuration( eutra_Cell1, l2Timer, v_Duration, float2int(p_RLC_Rec.t_Reordering * 1000.0) ) )

      {

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3A");

      }

    else

      {

        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 3A");

      }

    //@siclog "Step 3B" siclog@

    // The SS transmits RLC PDU#2 containing last segment of RLC SDU#1.

    p_RLC_Rec.UM_VTUS  := 1;

    f_TxUMD_PDUpartofSDU (p_RLC_Rec,

                          cs_TimingInfo_Now,

                          tsc_RLC_SDU1, 75, v_RLC_SDUsize - 75, tsc_FI_EndOfSDU);

    //@siclog "Step 4" siclog@

    // Check: Does the UE transmit RLC SDU#1 within 100 ms?

    t_WatchDog.start;

    alt

      {

        [] DRB.receive (car_DRB2_RLCUmDataInd (eutra_Cell1, ?))

          {

            t_WatchDog.stop;

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");

          }

        [] t_WatchDog.timeout

          {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

          }

      }

  } // fl_TC_7_2_2_8_TestBody

After change

 import from Parameters all;
 function fl_TC_7_2_2_8_TestBody(inout RLC_SS_State p_RLC_Rec) runs on EUTRA_PTC

  {

    var integer v_RLC_SDUsize := 128;

    var DRB_COMMON_IND v_ReceivedAsp;

    var RLC_DataField_Type v_SDU2_Received;

    var RLC_DataField_Type v_RLC_Data1;

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

    var SubFrameTiming_Type v_Timing, v_SubFrameTimingT1, v_SubFrameTimingT2;

    var integer v_SubframeOffset := 0;

    var integer v_Duration;

    timer t_WatchDog := 0.1;

    var integer v_Part1size := 75;    

    // determine RAT to make the timing correct later on

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    if (pc_ue_CategoryDL_M1){

      v_RLC_SDUsize := 100;

      v_Part1size := 60;

    }
    fl_GenerateRLC_UM_SDUs (p_RLC_Rec, v_RLC_SDUsize, 2);

    v_RLC_Data1 := substr(p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2],2,(v_RLC_SDUsize-2));

    v_SDU2_Received := f_PDCP_PDU_Encvalue (cs_PDCP_DataLongSN(0,v_RLC_Data1));

    //@siclog "Step 1" siclog@

    // The SS transmits RLC PDU#1 containing first segment of RLC SDU#1.

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    // Note T1

    v_SubFrameTimingT1.HSFN.Number := v_Timing.HSFN.Number;

    v_SubFrameTimingT1.SFN.Number := v_Timing.SFN.Number;

    v_SubFrameTimingT1.Subframe.Number := v_SubframeOffset;

    f_TxUMD_PDUpartofSDU (p_RLC_Rec,

                          cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number),

                          tsc_RLC_SDU1, 0, v_Part1size, tsc_FI_StartOfSDU);       

    //@siclog "Step 2" siclog@

    // The SS does not transmit RLC PDU#2 containing last segment of RLC SDU#1.

    // nothing sent in next subframe

    //@siclog "Step 3" siclog@

    // The SS transmits RLC PDU#3 containing RLC SDU#2.

    p_RLC_Rec.UM_VTUS  := 2;

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

    f_TxUMD_oneSDU( p_RLC_Rec,

                    cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number),

                    tsc_RLC_SDU2 );

    //@siclog "Step 3A" siclog@

    // Check 1: Does the UE transmit RLC SDU#2 after t-Reordering (200 ms) expiry?

    // Note T2.

    DRB.receive (car_DRB2_RLCUmDataInd (eutra_Cell1,

                                        cr_UMD_PDUoneSDU_SN10(p_RLC_Rec.UM_VRUR,

                                                              {v_SDU2_Received} ))) -> value v_ReceivedAsp;

    // Note T2.

    v_SubFrameTimingT2 := v_ReceivedAsp.Common.TimingInfo.SubFrame;

    v_Duration := f_EUTRA_NB_SubFrameTimingDuration( v_SubFrameTimingT1, v_SubFrameTimingT2);

    // Check 2: Is (T2-T1) > t-Reordering?

    if ( f_EUTRA_SubFrameTimingCheckDuration( eutra_Cell1, l2Timer, v_Duration, float2int(p_RLC_Rec.t_Reordering * 1000.0) ) )

      {

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3A");

      }

    else

      {

        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 3A");

      }

    //@siclog "Step 3B" siclog@

    // The SS transmits RLC PDU#2 containing last segment of RLC SDU#1.

    p_RLC_Rec.UM_VTUS  := 1;

    f_TxUMD_PDUpartofSDU (p_RLC_Rec,

                          cs_TimingInfo_Now,

                          tsc_RLC_SDU1, v_Part1size, v_RLC_SDUsize - v_Part1size, tsc_FI_EndOfSDU);   

    //@siclog "Step 4" siclog@

    // Check: Does the UE transmit RLC SDU#1 within 100 ms?

    t_WatchDog.start;

    alt

      {

        [] DRB.receive (car_DRB2_RLCUmDataInd (eutra_Cell1, ?))

          {

            t_WatchDog.stop;

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");

          }

        [] t_WatchDog.timeout

          {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");

          }

      }

  } // fl_TC_7_2_2_8_TestBody

Change 3 

	Function name
	fl_EUTRA_7_2_3_14_TestBody

	Reason for change
	The TC 7.2.3.14 has SDU sizes and combinations of SDU sizes that exceed the limitation for UL/DL grant for Cat-M1 UEs. The SDU sizes were modified to fit in 1000 bit TBS for Cat-M1 UEs.

	Summary of change
	Modified the SDU sizes for Cat-M1 UEs.

	TTCN module
	\LTE\7_2\RLC_AM_Testcases.ttcn

	MCC160 Comment
	


Before change

function fl_EUTRA_7_2_3_14_TestBody(inout RLC_SS_State p_RLC_Rec)

    runs on EUTRA_PTC

  { // @sic R5s100002 sic@ @sic R5s100279 sic@

    var integer i;

    var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

    var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

    var template (present) RLC_Status_Padding_Type v_Padding_1bit_Any := '?'B;

    var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

    var SubFrameTiming_Type v_Timing;

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

    var integer v_SubframeOffset;

    // SDU sizes employed: 100, 40, 130, 80, 150, 90, 200, 120, 60

    var integer v_PDCP_SDU_SizeArray[9]:= {100, 40, 130, 80, 150, 90, 200, 120, 60};

    timer t_WatchDog := 5.0;

    // determine RAT to make the timing correct later on

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    //PDCP SDUs of this size will be built, appended to a PDCP header,

    //and then used as RLC SDUs

    // generate 9 RLC SDUs = PDCP PDUs

    for (i := 0; i < 9; i:= i + 1) {

      p_RLC_Rec.TxDataSize := v_PDCP_SDU_SizeArray[i];

      f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

      v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

      p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

      p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

    }

    //@siclog "Step 1" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#1.

    f_TxAMD_SDU(p_RLC_Rec, cs_TimingInfo_Now, tsc_P_NoPoll, tsc_RLC_SDU1);

    //@siclog "Step 2" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#1.

    p_RLC_Rec.AM_VTS := p_RLC_Rec.AM_VTS - 1;

    f_TxAMD_SDU(p_RLC_Rec, cs_TimingInfo_Now, tsc_P_NoPoll, tsc_RLC_SDU1);

    //@siclog "Step 3" siclog@

    // Check: does the UE transmit RLC SDU#1?

    f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    //@siclog "Step 3A" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 4" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains SDU#2, and the 1st  part of SDU#3.

    // 40 + 2 respectively 80 + 2 octets

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 42);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 82);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_Now,

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 5" siclog@

    // Check: does the UE transmit RLC SDU#2?

    f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU2);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    //@siclog "Step 5A" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 6" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains SDU#2, and the 1st  part of SDU#3.

    // send PDU established in step 4 once again

    DRB.send (cas_DRB1_RLCAmDataReq (eutra_Cell1,

                                     cs_TimingInfo_Now,

                                     p_RLC_Rec.AMD_PDU));

    //@siclog "Step 7" siclog@

    // Check: does the UE transmit RLC SDU#2?

    t_WatchDog.start;

    alt {

      [] t_WatchDog.timeout

        {

        }

      [] DRB.receive (car_DRB1_RLCAmDataInd(eutra_Cell1, ?)) // catches more than SDU2

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 7");

        }

    } // alt

    //@siclog "Step 8" siclog@

    // The SS transmits an AMD PDU to the UE. This PDU contains the 2nd part of SDU#3.

    // This PDU contains the 2nd part of SDU#3.

    // 50 octets

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 82, 50);

    f_TxAMD_OnePDU(p_RLC_Rec,

                   cs_TimingInfo_Now,

                   tsc_P_NoPoll,

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1);

    //@siclog "Step 9" siclog@

    // Check: does the UE transmit RLC SDU#3?

    f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU3);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    //@siclog "Step 9A" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 11A" siclog@

    // The SS does not allocate any uplink grant.

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);  // @sic R5s100566 sic@

    // 80 + 2 respectively 60 + 2 octets

    //100 ms to wait for having an accurate reference to start

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    //@siclog "Step 10" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains the last part of SDU#6.

    // 70 octets

    p_RLC_Rec.AM_VTS := 5; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 22, 70);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    f_TxAMD_OnePDU(p_RLC_Rec,

                   cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                   tsc_P_NoPoll,

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1);

    //@siclog "Step 11" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains the 2nd part of SDU#5, and the 1st  part of SDU#6.

    // 2nd part of SDU#5 !!!

    // 90 respectively 20 + 2 octets

    p_RLC_Rec.AM_VTS := 4; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 62, 90);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 22);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                     tsc_P_NoPoll,

                     tsc_FI_MiddleOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 12" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#4 and the 1st part of SDU#5.

    p_RLC_Rec.AM_VTS := 3; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 82);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 62);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 12A" siclog@

    // The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback SDU#4, SDU#5 and SDU#6.

    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1,

                                        cs_TimingInfo(v_Timing));

    //@siclog "Step 13" siclog@

    // Check: Does the UE transmit an AMD PDU containing RLC SDU#4, RLC SDU#5 and RLC SDU#6 in its data field?

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 82);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 152);

    v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 92);

    f_RxAMD_PDU (p_RLC_Rec,

                 tsc_P_Poll,

                 tsc_FI_FullSDU,

                 cr_FlexPart_2LIs (82, 152),

                 {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3});

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");

    //@siclog "Step 14" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 16" siclog@

    // The SS transmits an AMD RLC PDU to the UE.

    // This PDU contains the last part of SDU#9.

    p_RLC_Rec.AM_VTS := 8; // SN to be used for the next PDU

    p_RLC_Rec.AM_VTA := 9; // set next SN the UE is to expect

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 32, 30);

    f_TxAMD_OnePDU(p_RLC_Rec,

                   cs_TimingInfo_Now,

                   tsc_P_Poll, // @sic R5s100172 sic@

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1);

    //@siclog "Step 17" siclog@

    // Check: Does the UE transmit a STATUS PDU NACK_SN/E1/E2 fields set correctly

    // to inform SS of missing PDUs #7, #8, (ACK_SN = 9, NACK_SN = 6, NACK_SN = 7)?

    v_NACK_List := cr_NACK_List_2NACKs(6,

                                       tsc_E2_NoSO_Follows,

                                       omit,

                                       7,

                                       tsc_E2_NoSO_Follows,

                                       omit);

    f_RxSTATUS_PDU_NACK(p_RLC_Rec,

                        v_NACK_List,

                        v_Padding_1bit_Any);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    //@siclog "Step 18" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains SDU#8, and the 1st  part of SDU#9.

    // 120 + 2 respectively 30 + 2 octets

    //100 ms to wait for having an accurate reference to start

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    p_RLC_Rec.AM_VTS := 7; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 122);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 32);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 18A" siclog@

    // The SS does not allocate any uplink grant.

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

    //@siclog "Step 19" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#7.

    // 200 + 2 octets

    p_RLC_Rec.AM_VTS := 6; // SN to be used for the next PDU

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

    f_TxAMD_SDU(p_RLC_Rec,

                cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                tsc_P_NoPoll,

                tsc_RLC_SDU7);

    //@siclog "Step 19A" siclog@

    // The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback SDU#7, SDU#8 and SDU#9.

    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1,

                                        cs_TimingInfo(v_Timing));

    //@siclog "Step 20" siclog@

    // Check: Does the UE transmit an AMD PDU containing RLC SDU#7, RLC SDU#8 and RLC SDU#9 in its data field?

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 202);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 122);

    v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 62);

    f_RxAMD_PDU (p_RLC_Rec,

                 tsc_P_Poll,

                 tsc_FI_FullSDU,

                 cr_FlexPart_2LIs (202, 122),

                 {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3});

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 20");

    //@siclog "Step 21" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

  } // fl_EUTRA_7_2_3_14_TestBody

After change

 import from Parameters all;
 function fl_EUTRA_7_2_3_14_TestBody(inout RLC_SS_State p_RLC_Rec)

    runs on EUTRA_PTC

  { // @sic R5s100002 sic@ @sic R5s100279 sic@

    var integer i;

    var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

    var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

    var template (present) RLC_Status_Padding_Type v_Padding_1bit_Any := '?'B;

    var template (value) RLC_Status_NACK_List_Type   v_NACK_List;

    var SubFrameTiming_Type v_Timing;

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

    var integer v_SubframeOffset;

    var integer v_1stPartSDU3 := 82;        

    var integer v_2ndPartSDU3 := 50;

    var integer v_1stPartSDU5 := 62;

    var integer v_2ndPartSDU5 := 90;

    var integer v_1stPartSDU6 := 22;

    var integer v_2ndPartSDU6 := 70;

    var integer v_1stPartSDU9 := 32;

    var integer v_2ndPartSDU9 := 30;
    // SDU sizes employed: 100, 40, 130, 80, 150, 90, 200, 120, 60

    var integer v_PDCP_SDU_SizeArray[9]:= {100, 40, 130, 80, 150, 90, 200, 120, 60};

    timer t_WatchDog := 5.0;

    if (pc_ue_CategoryDL_M1) {               

      v_PDCP_SDU_SizeArray := {100, 40, 100, 30, 30, 40, 30, 30, 40};

      v_1stPartSDU3 := 62;       

      v_2ndPartSDU3 := 40;

      v_1stPartSDU5 := 22;

      v_2ndPartSDU5 := 10;

      v_1stPartSDU6 := 22;

      v_2ndPartSDU6 := 20;

      v_1stPartSDU9 := 22;

      v_2ndPartSDU9 := 20;

    }
    // determine RAT to make the timing correct later on

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    //PDCP SDUs of this size will be built, appended to a PDCP header,

    //and then used as RLC SDUs

    // generate 9 RLC SDUs = PDCP PDUs

    for (i := 0; i < 9; i:= i + 1) {

      p_RLC_Rec.TxDataSize := v_PDCP_SDU_SizeArray[i];

      f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

      v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

      p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

      p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

    }

    //@siclog "Step 1" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#1.

    f_TxAMD_SDU(p_RLC_Rec, cs_TimingInfo_Now, tsc_P_NoPoll, tsc_RLC_SDU1);

    //@siclog "Step 2" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#1.

    p_RLC_Rec.AM_VTS := p_RLC_Rec.AM_VTS - 1;

    f_TxAMD_SDU(p_RLC_Rec, cs_TimingInfo_Now, tsc_P_NoPoll, tsc_RLC_SDU1);

    //@siclog "Step 3" siclog@

    // Check: does the UE transmit RLC SDU#1?

    f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    //@siclog "Step 3A" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 4" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains SDU#2, and the 1st  part of SDU#3.

    // 40 + 2 respectively 80 + 2 octets

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU2]+2);    

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, v_1stPartSDU3);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_Now,

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 5" siclog@

    // Check: does the UE transmit RLC SDU#2?

    f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU2);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    //@siclog "Step 5A" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 6" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains SDU#2, and the 1st  part of SDU#3.

    // send PDU established in step 4 once again

    DRB.send (cas_DRB1_RLCAmDataReq (eutra_Cell1,

                                     cs_TimingInfo_Now,

                                     p_RLC_Rec.AMD_PDU));

    //@siclog "Step 7" siclog@

    // Check: does the UE transmit RLC SDU#2?

    t_WatchDog.start;

    alt {

      [] t_WatchDog.timeout

        {

        }

      [] DRB.receive (car_DRB1_RLCAmDataInd(eutra_Cell1, ?)) // catches more than SDU2

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 7");

        }

    } // alt

    //@siclog "Step 8" siclog@

    // The SS transmits an AMD PDU to the UE. This PDU contains the 2nd part of SDU#3.

    // This PDU contains the 2nd part of SDU#3.

    // 50 octets

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], v_1stPartSDU3, v_2ndPartSDU3);

    f_TxAMD_OnePDU(p_RLC_Rec,

                   cs_TimingInfo_Now,

                   tsc_P_NoPoll,

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1);

    //@siclog "Step 9" siclog@

    // Check: does the UE transmit RLC SDU#3?

    f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU3);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    //@siclog "Step 9A" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 11A" siclog@

    // The SS does not allocate any uplink grant.

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);  // @sic R5s100566 sic@

    // 80 + 2 respectively 60 + 2 octets

    //100 ms to wait for having an accurate reference to start

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    //@siclog "Step 10" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains the last part of SDU#6.

    // 70 octets

    p_RLC_Rec.AM_VTS := 5; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], v_1stPartSDU6, v_2ndPartSDU6);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    f_TxAMD_OnePDU(p_RLC_Rec,

                   cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                   tsc_P_NoPoll,

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1);

    //@siclog "Step 11" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains the 2nd part of SDU#5, and the 1st  part of SDU#6.

    // 2nd part of SDU#5 !!!

    // 90 respectively 20 + 2 octets

    p_RLC_Rec.AM_VTS := 4; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], v_1stPartSDU5, v_2ndPartSDU5);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, v_1stPartSDU6);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                     tsc_P_NoPoll,

                     tsc_FI_MiddleOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 12" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#4 and the 1st part of SDU#5.

    p_RLC_Rec.AM_VTS := 3; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU4]+2);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, v_1stPartSDU5);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 12A" siclog@

    // The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback SDU#4, SDU#5 and SDU#6.

    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1,

                                        cs_TimingInfo(v_Timing));

    //@siclog "Step 13" siclog@

    // Check: Does the UE transmit an AMD PDU containing RLC SDU#4, RLC SDU#5 and RLC SDU#6 in its data field?

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU4]+2);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU5]+2);

    v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU6]+2);

    f_RxAMD_PDU (p_RLC_Rec,

                 tsc_P_Poll,

                 tsc_FI_FullSDU,

                 cr_FlexPart_2LIs (v_PDCP_SDU_SizeArray[tsc_RLC_SDU4]+2, v_PDCP_SDU_SizeArray[tsc_RLC_SDU5]+2),

                 {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3});

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");

    //@siclog "Step 14" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

    //@siclog "Step 16" siclog@

    // The SS transmits an AMD RLC PDU to the UE.

    // This PDU contains the last part of SDU#9.

    p_RLC_Rec.AM_VTS := 8; // SN to be used for the next PDU

    p_RLC_Rec.AM_VTA := 9; // set next SN the UE is to expect

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], v_1stPartSDU9, v_2ndPartSDU9);

    f_TxAMD_OnePDU(p_RLC_Rec,

                   cs_TimingInfo_Now,

                   tsc_P_Poll, // @sic R5s100172 sic@

                   tsc_FI_EndOfSDU,

                   v_RLC_Data1);

    //@siclog "Step 17" siclog@

    // Check: Does the UE transmit a STATUS PDU NACK_SN/E1/E2 fields set correctly

    // to inform SS of missing PDUs #7, #8, (ACK_SN = 9, NACK_SN = 6, NACK_SN = 7)?

    v_NACK_List := cr_NACK_List_2NACKs(6,

                                       tsc_E2_NoSO_Follows,

                                       omit,

                                       7,

                                       tsc_E2_NoSO_Follows,

                                       omit);

    f_RxSTATUS_PDU_NACK(p_RLC_Rec,

                        v_NACK_List,

                        v_Padding_1bit_Any);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    //@siclog "Step 18" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU contains SDU#8, and the 1st  part of SDU#9.

    // 120 + 2 respectively 30 + 2 octets

    //100 ms to wait for having an accurate reference to start

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    p_RLC_Rec.AM_VTS := 7; // SN to be used for the next PDU

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU8]+2);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, v_1stPartSDU9);

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 0);

    f_TxAMD_PDU_List(p_RLC_Rec,

                     cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                     tsc_P_NoPoll,

                     tsc_FI_StartOfSDU,

                     {v_RLC_Data1, v_RLC_Data2});

    //@siclog "Step 18A" siclog@

    // The SS does not allocate any uplink grant.

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

    //@siclog "Step 19" siclog@

    // The SS transmits an AMD PDU to the UE.

    // This PDU carries SDU#7.

    // 200 + 2 octets

    p_RLC_Rec.AM_VTS := 6; // SN to be used for the next PDU

    v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

    f_TxAMD_SDU(p_RLC_Rec,

                cs_TimingInfo_SubFrameOffset(v_Timing.SFN.Number, v_SubframeOffset, v_Timing.HSFN.Number), // @sic R5w160204r1 sic@

                tsc_P_NoPoll,

                tsc_RLC_SDU7);

    //@siclog "Step 19A" siclog@

    // The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback SDU#7, SDU#8 and SDU#9.

    v_Timing := f_EUTRA_NB_TimingInfoAdd(v_Timing, 60);

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1,

                                        cs_TimingInfo(v_Timing));

    //@siclog "Step 20" siclog@

    // Check: Does the UE transmit an AMD PDU containing RLC SDU#7, RLC SDU#8 and RLC SDU#9 in its data field?

    v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU7]+2);

    v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU8]+2);

    v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, v_PDCP_SDU_SizeArray[tsc_RLC_SDU9]+2);

    f_RxAMD_PDU (p_RLC_Rec,

                 tsc_P_Poll,

                 tsc_FI_FullSDU,

                 cr_FlexPart_2LIs (v_PDCP_SDU_SizeArray[tsc_RLC_SDU7]+2, v_PDCP_SDU_SizeArray[tsc_RLC_SDU7]+2),

                 {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3});

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 20");

    //@siclog "Step 21" siclog@

    // SS sends STATUS PDU upon Pollbit.

    f_TxSTATUS_PDU(p_RLC_Rec);

  } // fl_EUTRA_7_2_3_14_TestBody

