
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2016-06_D17wk10’.

2. Corrections required

2.1 Change 1 – Correction to function ‘f_NBIOT_MAC_CPMode_SendRlcStatus’

	Function name
	f_NBIOT_MAC_CPMode_SendRlcStatus()

	Reason for change
	When DL Scheduling assignments is set to DciDlInfo_Explicit in previous step of a test case and TTCN need to take care all DL MAC PDUs to match DL TB size.  Parameter p_DL_TBS has to be passed in this function.

	Summary of change
	Pass parameter p_DL_TBS for f_NBIOT_MAC_CPMode_SendRlcStatus.

	TTCN module
	NBIOT_MAC_Common.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_MAC_CPMode_SendRlcStatus(NB_RLC_CountsInfoList_Type p_RLC_CountsInfoList,

                                            NBIOT_CellId_Type p_CellId,

                                            template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now) runs on NBIOT_PTC

  {

    var integer v_AckSN := p_RLC_CountsInfoList.AM_VRR;

    var octetstring v_EncodedRlcPdu := f_RLC_Status_1PDU_Encvalue(v_AckSN);

    var MAC_SDUList_Type v_MAC_SDUList := { v_EncodedRlcPdu };

    f_NBIOT_MAC_CPMode_SendMacPdu(p_CellId, p_TimingInfo, f_NBIOT_MAC_PDU_WithSDUs_TX(tsc_LchId_SIB1bis, v_MAC_SDUList));

  }
After change

  function f_NBIOT_MAC_CPMode_SendRlcStatus(NB_RLC_CountsInfoList_Type p_RLC_CountsInfoList,

                                            NBIOT_CellId_Type p_CellId,

                                            template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                            template (omit) integer p_DL_TBS := omit) runs on NBIOT_PTC
  {

    var integer v_AckSN := p_RLC_CountsInfoList.AM_VRR;

    var octetstring v_EncodedRlcPdu := f_RLC_Status_1PDU_Encvalue(v_AckSN);

    var MAC_SDUList_Type v_MAC_SDUList := { v_EncodedRlcPdu };

    f_NBIOT_MAC_CPMode_SendMacPdu(p_CellId, p_TimingInfo, f_NBIOT_MAC_PDU_WithSDUs_TX(tsc_LchId_SIB1bis, v_MAC_SDUList, p_DL_TBS));

  }
2.2 Change 2 – Correction to function ‘f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX’

	Function name
	f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX()

	Reason for change
	If v_NoOfSDUs of p_MacSduList is 0, SduList should be set to omit when sending a MAC PDU.

	Summary of change
	If v_NoOfSDUs of p_MacSduList is 0, SduList should be set to omit when sending a MAC PDU.

	TTCN module
	NBIOT_MAC_Common.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(B5_Type p_LCID := tsc_LchId_SIB1bis,

                                                 NBIOT_MacPdu_PaddingType_Type p_PaddingType,

                                                 NBIOT_MacPdu_BSR_Type p_BSR,

                                                 template (present) MAC_SDUList_Type p_MacSduList := {}) return template (present) MAC_PDU_Type

  { /* build receive template of MAC PDU containing specific kind of padding (or no padding at all), short BSR or no BSR and 0, 1 or several MAC SDUs

       NOTE: Due to the fact that there is only one logical channel there should be just one RLC AM PDU (if any) i.e. a maximum of one MAC SDU */

    var template (present) MAC_Header_Type v_MacPduHeader := {};

    var template MAC_CTRL_ElementList_Type v_CtrlElementList := omit;

    var template octetstring v_PaddingAtEndOfPDU := omit;

    var integer v_NoOfSDUs := lengthof(p_MacSduList);

    var B1_Type v_ExtensionBit := tsc_ExtensionMoreHeaders;

    var template (omit) integer v_Length;

    var integer i;

    select (p_PaddingType) {

      case (noPadding) {

      }

      case (singleBytePadding) {

        v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

      }

      case (twoBytePadding) {

        v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

        v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

      }

      case (normalPadding) {

        v_PaddingAtEndOfPDU := ?;

      }

    }

    if (p_BSR != noBSR) {

      if ((v_NoOfSDUs == 0) and not ispresent(v_PaddingAtEndOfPDU)) {

        v_ExtensionBit := tsc_ExtensionNoMoreHeaders;

      }

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_ShortBSR, v_ExtensionBit);

      if (p_BSR == zeroBSR) {

        v_CtrlElementList := cr_NBIOT_MAC_CTRL_ElementList_ZeroBSR;

      } else {

        v_CtrlElementList := cr_NBIOT_MAC_CTRL_ElementList_AnyBSR;

      }

    }

    for (i := 0; i < v_NoOfSDUs; i := i + 1) {

      if (not ispresent(v_PaddingAtEndOfPDU) and (i == v_NoOfSDUs - 1)) {

        v_ExtensionBit := tsc_ExtensionNoMoreHeaders;

        v_Length := omit;

      } else {

        v_Length := lengthof(p_MacSduList[i]);

      }

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, p_LCID, v_ExtensionBit, v_Length);

    }

    if (ispresent(v_PaddingAtEndOfPDU)) {

      v_ExtensionBit := tsc_ExtensionNoMoreHeaders;

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

    }

    return cr_MAC_PDU(v_MacPduHeader, v_CtrlElementList, p_MacSduList, v_PaddingAtEndOfPDU);

  }
After change

  function f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(B5_Type p_LCID := tsc_LchId_SIB1bis,

                                                 NBIOT_MacPdu_PaddingType_Type p_PaddingType,

                                                 NBIOT_MacPdu_BSR_Type p_BSR,

                                                 template (present) MAC_SDUList_Type p_MacSduList := {}) return template (present) MAC_PDU_Type

  { /* build receive template of MAC PDU containing specific kind of padding (or no padding at all), short BSR or no BSR and 0, 1 or several MAC SDUs

       NOTE: Due to the fact that there is only one logical channel there should be just one RLC AM PDU (if any) i.e. a maximum of one MAC SDU */

    var template (present) MAC_Header_Type v_MacPduHeader := {};

    var template MAC_CTRL_ElementList_Type v_CtrlElementList := omit;

    var template octetstring v_PaddingAtEndOfPDU := omit;

    var integer v_NoOfSDUs := lengthof(p_MacSduList);

    var B1_Type v_ExtensionBit := tsc_ExtensionMoreHeaders;

    var template (omit) integer v_Length;

    var integer i;

    select (p_PaddingType) {

      case (noPadding) {

      }

      case (singleBytePadding) {

        v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

      }

      case (twoBytePadding) {

        v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

        v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

      }

      case (normalPadding) {

        v_PaddingAtEndOfPDU := ?;

      }

    }

    if (p_BSR != noBSR) {

      if ((v_NoOfSDUs == 0) and not ispresent(v_PaddingAtEndOfPDU)) {

        v_ExtensionBit := tsc_ExtensionNoMoreHeaders;

      }

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_ShortBSR, v_ExtensionBit);

      if (p_BSR == zeroBSR) {

        v_CtrlElementList := cr_NBIOT_MAC_CTRL_ElementList_ZeroBSR;

      } else {

        v_CtrlElementList := cr_NBIOT_MAC_CTRL_ElementList_AnyBSR;

      }

    }

    for (i := 0; i < v_NoOfSDUs; i := i + 1) {

      if (not ispresent(v_PaddingAtEndOfPDU) and (i == v_NoOfSDUs - 1)) {

        v_ExtensionBit := tsc_ExtensionNoMoreHeaders;

        v_Length := omit;

      } else {

        v_Length := lengthof(p_MacSduList[i]);

      }

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, p_LCID, v_ExtensionBit, v_Length);

    }

    if (ispresent(v_PaddingAtEndOfPDU)) {

      v_ExtensionBit := tsc_ExtensionNoMoreHeaders;

      v_MacPduHeader := f_NBIOT_MacPduHeader_AddSubheader_RX(v_MacPduHeader, tsc_LchId_Padding, v_ExtensionBit);

}
    if(0 == v_NoOfSDUs)
    {
      return cr_MAC_PDU(v_MacPduHeader, v_CtrlElementList, omit, v_PaddingAtEndOfPDU);
    }

    else

    {
      return cr_MAC_PDU(v_MacPduHeader, v_CtrlElementList, p_MacSduList, v_PaddingAtEndOfPDU);

    }
  }
2.3 Change 3 – Correction to function ‘f_NBIOT_MacSdu_RlcAmdPdu_RX’

	Function name
	f_NBIOT_MacSdu_RlcAmdPdu_RX()

	Reason for change
	1. When check UL data is matched with received template or not, TTCN shall encode expected UL data at first. So create new function f_RLC_AMD_PDU_NoSeg_Encvalue to encode 1-3 RLC UL PDUs data(enough for all MAC test case by now).
2. When encode expected UL data with 2 RLC PDUs in f_RLC_AMD_PDU_NoSeg_Encvalue, padding part of RLC AMD PDU header  (with 1 LI 11 Length Indicator) can not be ‘?’, so pass parameter p_Padding for f_RLC_AMD_PDU_NoSeg_Encvalue and f_NBIOT_MacSdu_RlcAmdPdu_RX, and create new template cr_FlexPart_1LI_Padding.
3. Pass inout parameter p_RLC_CountsInfoLis to generate expected UL data.

	Summary of change
	1. Create new function f_RLC_AMD_PDU_NoSeg_Encvalue to encode 1-3 RLC UL PDUs data.
2. Pass parameter p_Padding for f_RLC_AMD_PDU_NoSeg_Encvalue and f_NBIOT_MacSdu_RlcAmdPdu_RX, and create new template cr_FlexPart_1LI_Padding. 

3. Pass inout parameter p_RLC_CountsInfoLis to generate expected UL data.

	TTCN module
	NBIOT_MAC_Common.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_MacSdu_RlcAmdPdu_RX(template (value) RLC_SDUList_Type p_RlcSduList) return template (present) octetstring

  { /* in NBIOT for CP mode all RLC SDUs are on the same logical channel i.e. send in one RLC PDU by the UE

       =>  for a single RLC SDU the AM PDU has 2 bytes fixed header part only, but for more than one RLC SDUs there is a flex header part with LIs for the first N-1 SDUs

       function returns an octetstring with any content but with the fix length of the expected AM PDU

       => the content is not checked (could be enhanced in the future if needed) */

    var integer v_NoOfSDUs := lengthof(p_RlcSduList);

    var integer v_NoOfLIs := v_NoOfSDUs - 1;                    /* LIs for all SDUs but not the last one */

    var integer v_FixPartLength := 2;                           /* 2 bytes for RLC AM PDU header fixed part */

    var integer v_FlexPartLength := (12 * v_NoOfLIs + 7) / 8;   /* see 36.322 clause 6.2.1.4 */

    var integer v_RlcAmHeaderLength := v_FixPartLength + v_FlexPartLength;

    var integer v_RlcAmPduLength := v_RlcAmHeaderLength;

    var template (present) octetstring v_RlcAmPdu;

    var integer i;

    for (i := 0; i < v_NoOfSDUs; i := i + 1) {

      v_RlcAmPduLength := v_RlcAmPduLength + lengthof(p_RlcSduList[i]);

    }

    v_RlcAmPdu := ? length(v_RlcAmPduLength);

/*     !!!! Does not work with Titan !!!! */

/*     v_RlcAmPdu := ? length(v_RlcAmHeaderLength); */

/*     for (i := 0; i < v_NoOfSDUs; i := i + 1) { */

/*       v_RlcAmPdu := v_RlcAmPdu & p_RlcSduList[i]; */

/*     } */
    return v_RlcAmPdu;

  }
After change

//New template for RLC AMD PDU header of 2 RLC PDUs with padding parameter
  template (value) RLC_PDU_Header_FlexPart_Type cr_FlexPart_1LI_Padding(integer p_LI,

                                                                B4_Type p_Padding := '0000'B) :=

  { // serves to transmit FlexPart with 1 LI 11 Length Indicator (AM)

    LengthIndicator := {

      LI11 := {

        cs_LI11(p_LI)

      }

    },

    Padding := p_Padding

  };
//New function for Encoding 1-3 RLC PDUs
  function f_RLC_AMD_PDU_NoSeg_Encvalue(integer p_SN,

                                        B1_Type p_Poll,

                                        template (value) RLC_DataFieldList_Type p_RLC_DataList,

                                        RLC_FramingInfo_Type p_FramingInfo:= tsc_FI_FullSDU,

                                        B4_Type p_FlexPart_Padding := '0000'B) return octetstring

  {

    var template (value) RLC_PDU_Type v_RLC_PDU;

    var template (value) RLC_AMD_PDU_Type v_AMD_PDU_NoSeg;

    var integer v_LI1, v_LI2;

    select (lengthof (p_RLC_DataList)) {

      case (1) {

        v_AMD_PDU_NoSeg :=  cs_AMD_PDU_LIs(p_SN, p_Poll, p_FramingInfo, omit, p_RLC_DataList);

      }

      case (2) {

        v_LI1 := lengthof (p_RLC_DataList[0]);

        v_AMD_PDU_NoSeg := cs_AMD_PDU_LIs (p_SN,             // Note: valueof cannot be avoided here

                                           p_Poll,

                                           p_FramingInfo,

                                           cr_FlexPart_1LI_Padding(v_LI1, p_FlexPart_Padding),

                                           p_RLC_DataList);

      }

      case (3) {

        v_LI1 := lengthof (p_RLC_DataList[0]);

        v_LI2 := lengthof (p_RLC_DataList[1]);

        v_AMD_PDU_NoSeg := cs_AMD_PDU_LIs (p_SN,             // Note: valueof cannot be avoided here

                                           p_Poll,

                                           p_FramingInfo,

                                           cs_FlexPart_2LIs(v_LI1, v_LI2),

                                           p_RLC_DataList);

      }

      case else {

        FatalError(__FILE__, __LINE__, "Function used in a currently unforeseen way");

      }

    }

    v_RLC_PDU := {

          AMD := v_AMD_PDU_NoSeg

        };

    return f_RLC_PDU_Encvalue(v_RLC_PDU);

  }
  function f_NBIOT_MacSdu_RlcAmdPdu_RX(inout NB_RLC_CountsInfoList_Type p_RLC_CountsInfoList,

                                       template (value) RLC_SDUList_Type p_RlcSduList,

                                       B4_Type p_FlexPart_Padding := '0000'B) return template (present) octetstring

  { /* in NBIOT for CP mode all RLC SDUs are on the same logical channel i.e. send in one RLC PDU by the UE

       =>  for a single RLC SDU the AM PDU has 2 bytes fixed header part only, but for more than one RLC SDUs there is a flex header part with LIs for the first N-1 SDUs

       function returns an octetstring with any content but with the fix length of the expected AM PDU

       => the content is not checked (could be enhanced in the future if needed) */

    var integer v_NoOfSDUs := lengthof(p_RlcSduList);

    var integer v_NoOfLIs := v_NoOfSDUs - 1;                    /* LIs for all SDUs but not the last one */

    var integer v_FixPartLength := 2;                           /* 2 bytes for RLC AM PDU header fixed part */

    var integer v_FlexPartLength := (12 * v_NoOfLIs + 7) / 8;   /* see 36.322 clause 6.2.1.4 */

    var integer v_RlcAmHeaderLength := v_FixPartLength + v_FlexPartLength;

    var integer v_RlcAmPduLength := v_RlcAmHeaderLength;

    var template (present) octetstring v_RlcAmPdu;

    var integer i;

    for (i := 0; i < v_NoOfSDUs; i := i + 1) {

      v_RlcAmPduLength := v_RlcAmPduLength + lengthof(p_RlcSduList[i]);

    }

    v_RlcAmPdu := f_RLC_AMD_PDU_NoSeg_Encvalue(p_RLC_CountsInfoList.AM_VRR, tsc_P_Poll, p_RlcSduList, -, p_FlexPart_Padding);
    return v_RlcAmPdu;

  }
2.4 Change 4 – Correction to function ‘f_NBIOT_MacPdu_SingleRlcAmdPdu_RX’

	Function name
	f_NBIOT_MacPdu_SingleRlcAmdPdu_RX()

	Reason for change
	Pass inout parameter p_RLC_CountsInfoLis to generate expected UL data.

	Summary of change
	Pass inout parameter p_RLC_CountsInfoLis to generate expected UL data.

	TTCN module
	NBIOT_MAC_Common.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_MacPdu_SingleRlcAmdPdu_RX(B5_Type p_LCID := tsc_LchId_SIB1bis,

                                             template (value) RLC_SDU_Type p_RlcSdu) return template (present) MAC_PDU_Type

  { /* build receive template of MAC PDU to be used to receive a single MAC SDU (RLC AM PDU) allowing any padding and any CTRL elements */

    var template (present) MAC_Header_Type v_Header := {*, cr_MAC_PDU_SubHeader_AnySDU(p_LCID), *};

    var template (value) RLC_SDUList_Type v_RlcSduList := { p_RlcSdu };

    var template (present) octetstring v_RlcAmPdu := f_NBIOT_MacSdu_RlcAmdPdu_RX(v_RlcSduList);

    var template (present) MAC_SDUList_Type v_SduList := { v_RlcAmPdu };

    return cr_MAC_PDU(v_Header, -, v_SduList);

  }
After change

  function f_NBIOT_MacPdu_SingleRlcAmdPdu_RX(inout NB_RLC_CountsInfoList_Type p_RLC_CountsInfoList,

                                             B5_Type p_LCID := tsc_LchId_SIB1bis,

                                             template (value) RLC_SDU_Type p_RlcSdu) return template (present) MAC_PDU_Type

  { /* build receive template of MAC PDU to be used to receive a single MAC SDU (RLC AM PDU) allowing any padding and any CTRL elements */

    var template (present) MAC_Header_Type v_Header := {*, cr_MAC_PDU_SubHeader_AnySDU(p_LCID), *};

    var template (value) RLC_SDUList_Type v_RlcSduList := { p_RlcSdu };

    var template (present) octetstring v_RlcAmPdu := f_NBIOT_MacSdu_RlcAmdPdu_RX(p_RLC_CountsInfoList, v_RlcSduList);

    var template (present) MAC_SDUList_Type v_SduList := { v_RlcAmPdu };

    return cr_MAC_PDU(v_Header, -, v_SduList);

  }
2.5 Change 5 – Correction to function ‘f_NBIOT_MAC_CPMode_RlcStatusPdu_RlcAmdPdu_Receive’

	Function name
	f_NBIOT_MAC_CPMode_RlcStatusPdu_RlcAmdPdu_Receive()

	Reason for change
	1.Since p_RLC_CountsInfoList is passed in f_NBIOT_MacSdu_RlcAmdPdu_RX above, add p_RLC_CountsInfoList in function.
2. ‘v_AnyRlcStatusPdu := ? length(2)’ will cause error when TTCN tools running at  ‘receive’ operation.

	Summary of change
	1. Add p_RLC_CountsInfoList in function.
2. Generate expected RLC Status PDU with ACK intead of ‘? length(2)’.

	TTCN module
	NBIOT_MAC_Common.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_MAC_CPMode_RlcStatusPdu_RlcAmdPdu_Receive(inout NB_RLC_CountsInfoList_Type p_RLC_CountsInfoList,

                                                             NBIOT_CellId_Type p_CellId,

                                                             template (value) RLC_SDUList_Type p_RlcSduList,

                                                             NBIOT_MacPdu_PaddingType_Type p_PaddingType := noPadding,

                                                             NBIOT_MacPdu_BSR_Type p_BSR := noBSR) runs on NBIOT_PTC

  {

    var template (present) octetstring v_RlcAmdPdu := f_NBIOT_MacSdu_RlcAmdPdu_RX(p_RlcSduList);

    var template (present) octetstring v_AnyRlcStatusPdu := ? length(2);
    var template (present) MAC_SDUList_Type v_MAC_SDUList_RX_1 := { v_RlcAmdPdu, v_AnyRlcStatusPdu };

    var template (present) MAC_SDUList_Type v_MAC_SDUList_RX_2 := { v_AnyRlcStatusPdu, v_RlcAmdPdu };

    var template (present) MAC_PDU_Type v_MAC_PDU_RX := (f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, p_PaddingType, p_BSR, v_MAC_SDUList_RX_1),

                                                         f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, p_PaddingType, p_BSR, v_MAC_SDUList_RX_2));

    f_NBIOT_MAC_CPMode_MacPdu_Receive(p_RLC_CountsInfoList, p_CellId, v_MAC_PDU_RX);

  }
After change

//New template for
  template RLC_PDU_Type cr_RLC_PDU_StatusPDU_ACK(integer p_AckSN) := {

    Status := cr_RLC_AM_StatusPDU_ACK_SN(p_AckSN)

  }
  function f_NBIOT_MAC_CPMode_RlcStatusPdu_RlcAmdPdu_Receive(inout NB_RLC_CountsInfoList_Type p_RLC_CountsInfoList,

                                                             NBIOT_CellId_Type p_CellId,

                                                             template (value) RLC_SDUList_Type p_RlcSduList,

                                                             NBIOT_MacPdu_PaddingType_Type p_PaddingType := noPadding,

                                                             NBIOT_MacPdu_BSR_Type p_BSR := noBSR) runs on NBIOT_PTC

  {

    var template (present) octetstring v_RlcAmdPdu := f_NBIOT_MacSdu_RlcAmdPdu_RX(p_RLC_CountsInfoList, p_RlcSduList);

    var bitstring v_Ack := int2bit(p_RLC_CountsInfoList.AM_VTS, 10);

    var template (present) O2_Type v_AnyRlcStatusPdu := (bit2oct('0000'B & v_Ack & '0'B & '0'B),bit2oct('0000'B & v_Ack & '0'B & '1'B));
    var template (present) MAC_SDUList_Type v_MAC_SDUList_RX_1 := { v_RlcAmdPdu, v_AnyRlcStatusPdu };

    var template (present) MAC_SDUList_Type v_MAC_SDUList_RX_2 := { v_AnyRlcStatusPdu, v_RlcAmdPdu };

    var template (present) MAC_PDU_Type v_MAC_PDU_RX := (f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, p_PaddingType, p_BSR, v_MAC_SDUList_RX_1),

                                                         f_NBIOT_MacPdu_BSR_MacSdus_Padding_RX(-, p_PaddingType, p_BSR, v_MAC_SDUList_RX_2));

    f_NBIOT_MAC_CPMode_MacPdu_Receive(p_RLC_CountsInfoList, p_CellId, v_MAC_PDU_RX);

  }
